
Programming the Windows Graphical User Interface - Part Two

Adding Menus to a Window

We've shown how to add buttons to windows so that the user can invoke actions.  WINDOWS also lets us add pull-down menus to our programs.  These include the common File, Edit, View, Help, etc. menus.  You can add dozens of user actions to a window with menus, but just try adding dozens of buttons to a window.  You wouldn't have room for anything else!

Just look at this simple example:

  'lunch today
  label$ = "Your selection: "
  menu #main, "&Lunch Specials", _
     "&Scrod soup", [scrod], _
     "&Haggis", [haggis], _
     "&Mud Pie", [mudPie]
  statictext #main.choice, label$, 10, 10, 250, 30
  open "May I help you?" for window as #main
  #main "trapclose [quit]"
  wait

[scrod]
  #main.choice label$ + "Scrod soup?  What's that?"
  wait

[haggis]
  #main.choice label$ + "Stuffed sheep stomach, Oy!"
  wait

[mudPie]
  #main.choice label$ + "Mud pie looks like fudge!"
  wait

[quit]
  close #main
  end

Notice that we use the underscore discussed in a previous lesson to break up the MENU command in our example.  Menus that have more than a couple of items tend to get pretty long, so using the underscore to break them up over several lines is common.

Just like the commands for adding other kinds of controls like buttons and textboxes, the MENU command must come before the window is opened.

The menu has a title 'Lunch Specials', and three items named Scrod, Haggis, and Mud Pie.  Each of these items has a branch label to jump to.  Also see the ampersand & characters.  These give the letter following them special accelerator status.  To invoke a menu item using accelerators, use the Alt key and the indicated letter in the menu title (Alt+L in the example).  This causes the menu to appear on screen.  Then find the desired menu action item and press the underline letter on the keyboard.  So for scrod soup, press Alt+L and then S.

You can have more than one menu.  Right after the first MENU command we could add a second menu to our example program that might look something like:

  menu #main, "&Food Alert", _
     "&Head for the hills", [headForHills], _
     "&Slip the food to the dog", [feedDog], _
     "&Feed parking meter excuse", [parkingMeter] _
     "&Dial 911", [dial911]

Another thing we can do it add separators between items in the menu.  This is a visual cue that groups related menu items together.  We do this by putting a | (on most keyboards this is Shift+\) between items in the menu command.  Try substituting the following menu command in the example program above.  Run it to see what this looks like:

  menu #main, "&Lunch Specials", _
     "&Scrod soup", [scrod], _
     |, _
     "&Haggis", [haggis], _
     "&Mud Pie", [mudPie]


Using NOMAINWIN

The default text window that opens with each Liberty BASIC program we create can be optionally turned off.  This can be desirable when we don't want our application to be operated out of a text window.  The command that accomplishes this effect is NOMAINWIN.  Simply include NOMAINWIN in your program's source code anywhere at all.  This accomplishes two things:

1) It hides the program's main window;
2) It prevents the 'Execution Complete' notice from appearing at the end of a program run.

This comes at a cost.  Your program must close itself down properly when all the windows have been closed.  Check out the example below.  

  'demonstrate the NOMAINWIN command
  'the window is sized at 300 by 100
  nomainwin  'do not open a main window
  WindowWidth = 300
  WindowHeight = 100
  button #example.ok, "OK!", [okClicked], UL, 220, 35
  textbox #example.field, 10, 35, 200, 25
  statictext #example.label, "Type something", 10, 10, 150, 25
  open "myname's first window!" for window as #example
  #example "trapclose [quit]"
  wait   'stop and wait for user interaction

[okClicked]  'Get the contents of the entry field
  'now get the contents from the textbox
  #example.field "!contents? aString$"
  'now pop up a notice saying what was in the textbox
  notice aString$
  wait   'stop and wait for user interaction

[quit]
  'ask if the user wants to quit
  confirm "Really Quit?"; answer$
  if answer$ <> "yes" then wait 'abort quitting
  'now close the window
  close #example
  end

After the CLOSE #example statement in our program above, we have an END statement.   If we didn't, then our program would not terminate properly, and there would be no error message to clue us in.  Always use END in your programs because just closing all the windows doesn't end the program.  There are kinds of programs which don't have windows open all the time, and even some which don't have any windows at all.

Note: If you hide the main window with NOMAINWIN, you can still PRINT to it.  The output will go into limbo (nowhere).  Even if your program doesn't need a main window, this can be a really handy way to log debugging commentary.  When the program runs to your satisfaction, put a NOMAINWIN command in there to hide the debugging log.  Then if the extra invisible PRINT commands make your program run too slowly, remove them.

The Structure of a Liberty BASIC GUI Program

We've taken a look at some simple elements of Liberty BASIC GUI programming.  Now let's consider how to factor code in a logical and practical way when creating GUI programs in LB.

Organization

In the case of a project that uses more than one window, we should make sure that code for each window is grouped together.  Most applications are designed around a cental window that is always open and from where other windows are launched.  When the central window is closed, the application ends.  Liberty BASIC is a good example of this kind of structure.  When you start it up the Liberty BASIC editor appears.  From there you can open some other windows for opening and saving files, for setting preferences and more.  Until you close the Liberty BASIC editor, Liberty BASIC continues to run.

Here is a simple outline (not a working program) that demonstrates this idea:

  'This is an example outline
  '--Define our main application window here
  button #main.1, ....
  button #main.2, ....
  textbox ....
  open "My Application" for window as #main
  #main "trapclose [closeMain]"
  wait  'stop and wait for user action

[button1Clicked]
  'do something here for button1
  wait

[button2Clicked]
  'do something here for button2
  wait

[closeMain]  'perform some cleanup code and close the window
  close #main
  gosub [sub1]  'act on user action

  'end application
  end

[openWin2]   '--Define our second window here
  button #2.3, ....
  button #2.4, ....
  textbox ....
  open "Window 2" for window as #2
  #2 "trapclose [close2]"
  wait  'stop and wait here for user action

[button3Clicked]
  'do something here for button3
  wait

[button4Clicked]
  'do something here for button4
  wait

[close2]  'perform some cleanup code and close the window
  close #2
  gosub [sub2]  'act on user action
  wait

'--------------------------------------------------
  'We might place code here containing
  'general subroutines


See how we arrange the code for each window like so:

 [openMainWindow]
 [userAction1]
 [userAction2]
 [closeMainWindow]
     end

 [openWindow1]
 [userAction3]
 [userAction4]
 [closeWindow1]

 [openWindow2]
 [userAction5]
 [userAction6]
 [closeWindow2]

The names we give to our branch labels above are of course generic.  All the code for a given window and user actions performed on it should be grouped together.  By structuring things in this way it is relatively easy to develop and debug your programs.

NOTE: Be careful to avoid creating a program that closes all its windows but doesn't actually end.  If the program stops at an input or wait statement after closing all of it's windows, it will not terminate properly and resources will not be freed.  If the program loops continuously after closing all its windows then  Liberty BASIC's WINDOWS message handler will be locked out.  In this case press CTRL-Break to halt the errant program, then use the Kill BASIC Programs item from the Liberty BASIC Run menu to clean up the program.

Using Dialog Boxes

Dialog boxes are a special kind of window.  They are different in several respects from other windows:

1) Dialog boxes can be made modal.  This means that if I have a first window open and then I open a modal dialog box, that I will be unable to select the first window until the modal dialog is closed.  Most 'About' boxes are modal dialogs, as are many 'Properties' type windows.  For an example, select Liberty BASIC's Help menu and click on 'About Liberty BASIC'.  You won't be able to do anything else with Liberty BASIC until you close the dialog box that appears.

2) Dialog boxes cannot be resized by the user, and they don't have minimize or maximize buttons.

3) Dialog boxes can have a default pushbutton which is activated when the use presses the Enter key.  Most of the time a dialog box's default pushbutton is the OK or Accept button.

4) Dialog boxes cannot have pull-down menus.

Some applications are appropriately constructed completely of dialog boxes and have no other kind of window at all.  Take a look at the program FreeForm which comes with Liberty BASIC.  This program is more typical.  The main application windows is a graphics type, but most of the other windows are dialog boxes.


More About Statictext

We saw in the first part of this lesson how a statictext control is used to label other controls in a window.  Statictext controls do not have to display the same label all the time.  We can print to them in similar fashion to the way we print to a textbox control.  Try the example code below:

  'demonstrate printing to statictext control
  nomainwin
  statictext #main.statictext, "Counting down:", 10, 10, 200, 25
  open "Countdown" for window as #main
  #main "trapclose [quit]"
  for count = 10 to 1 step -1
    #main.statictext "Counting down: "; count
    t$ = time$()
    while t$ = time$()  'loop until the time changes
    wend
  next count
  #main.statictext "Counting Down: Done."
  wait
[quit]
  close #main
  end

Since we can change the contents of a statictext control, it isn't really static.  Pesky little thing isn't it?  ;-)


Lesson Exercise - HILO Revisited

Create a version of the HILO program using the techniques in this lesson.  The program should have its own application window using the window type.  This window should have at least one statictext control.  One of these should indicate whether to guess higher or lower and to count the guesses.  There should be a textbox for entering guesses, and a button labeled 'Guess' to click on to register each guess.  An additional button should be labeled 'About'.  This button should cause an About box to be displayed for the game.  The About box should be constructed as a modal dialog box using an OPEN command similar to:

  'this would be just part of a larger program
  WindowWidth = 160
  WindowHeight = 100
  statictext #about, "I am an example about box", 10, 10, 150, 25
  button #about.ok, "OK", [closeAbout], UL,  50, 40, 50, 25
  open "My About Box" for dialog_modal as #about
  #about "trapclose [closeAbout]"
  wait
[closeAbout]
  close #about

In addition to the buttons, add a menu titled 'Play' with two action items called 'Guess' and 'Exit'.  Guess should perform the same action as clicking in the Guess button, and Exit should cause the game to quit.  Add a second menu titled 'Help' with one action item called 'About HILO' which performs the same action as clicking the About button.


Lesson Eight Part Two Quiz - Answers further down the page

1) Use the ampersand character to add _________________ to menus.

2) Dialog boxes can be made ____________, but regular windows cannot.

3) Users cannot ________________ dialog boxes.

4) To hide the main window, use the __________________ command.

5) When quitting your program, make sure to use an ____________ command

    after closing all the windows to make sure the program terminates properly.

6) To change the contents of a statictext control we ______________ to it.
































Lesson Eight Part Two Quiz Answers

1) accelerators

2) modal

3) resize

4) NOMAINWIN

5) END

6) PRINT


